NAG Library Routine Document

F07JBF (DPTSVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07JBF (DPTSVX) uses the factorization

$$A = LDL^{\mathrm{T}}$$

to compute the solution to a real system of linear equations

$$AX = B,$$

where A is an n by n symmetric positive definite tridiagonal matrix and X and B are n by r matrices. Error bounds on the solution and a condition estimate are also provided.

2 Specification

```
SUBROUTINE F07JBF (FACT, N, NRHS, D, E, DF, EF, B, LDB, X, LDX, RCOND,     &
                   FERR, BERR, WORK, INFO)
INTEGER            N, NRHS, LDB, LDX, INFO
REAL (KIND=nag_wp) D(*), E(*), DF(*), EF(*), B(LDB,*), X(LDX,*), RCOND,    &
                   FERR(NRHS), BERR(NRHS), WORK(2*N)
CHARACTER(1)       FACT
```

The routine may be called by its LAPACK name ***dptsvx***.

3 Description

F07JBF (DPTSVX) performs the following steps:

1. If FACT = 'N', the matrix A is factorized as $A = LDL^{\mathrm{T}}$, where L is a unit lower bidiagonal matrix and D is diagonal. The factorization can also be regarded as having the form $A = U^{\mathrm{T}}DU$.
2. If the leading i by i principal minor is not positive definite, then the routine returns with INFO $= i$. Otherwise, the factored form of A is used to estimate the condition number of the matrix A. If the reciprocal of the condition number is less than ***machine precision***, INFO $= \mathrm{N} + 1$ is returned as a warning, but the routine still goes on to solve for X and compute error bounds as described below.
3. The system of equations is solved for X using the factored form of A.
4. Iterative refinement is applied to improve the computed solution matrix and to calculate error bounds and backward error estimates for it.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

1: FACT – CHARACTER(1) *Input*

On entry: specifies whether or not the factorized form of the matrix A has been supplied.

FACT = 'F'
: DF and EF contain the factorized form of the matrix A. DF and EF will not be modified.

FACT = 'N'
: The matrix A will be copied to DF and EF and factorized.

Constraint: FACT = 'F' or 'N'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: $\mathrm{NRHS} \geq 0$.

4: D($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: the n diagonal elements of the tridiagonal matrix A.

5: E($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N}-1)$.

On entry: the $(n-1)$ subdiagonal elements of the tridiagonal matrix A.

6: DF($*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array DF must be at least $\max(1, \mathrm{N})$.

On entry: if FACT = 'F', DF must contain the n diagonal elements of the diagonal matrix D from the LDL^{T} factorization of A.

On exit: if FACT = 'N', DF contains the n diagonal elements of the diagonal matrix D from the LDL^{T} factorization of A.

7: EF($*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array EF must be at least $\max(1, \mathrm{N}-1)$.

On entry: if FACT = 'F', EF must contain the $(n-1)$ subdiagonal elements of the unit bidiagonal factor L from the LDL^{T} factorization of A.

On exit: if FACT = 'N', EF contains the $(n-1)$ subdiagonal elements of the unit bidiagonal factor L from the LDL^{T} factorization of A.

8: B(LDB, $*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

9: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07JBF (DPTSVX) is called.

Constraint: $\text{LDB} \geq \max(1, \text{N})$.

10: $\text{X}(\text{LDX}, *)$ – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array X must be at least $\max(1, \text{NRHS})$.

On exit: if $\text{INFO} = 0$ or $\text{N} + 1$, the n by r solution matrix X.

11: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which F07JBF (DPTSVX) is called.

Constraint: $\text{LDX} \geq \max(1, \text{N})$.

12: RCOND – REAL (KIND=nag_wp) *Output*

On exit: the reciprocal condition number of the matrix A. If RCOND is less than the ***machine precision*** (in particular, if $\text{RCOND} = 0.0$), the matrix is singular to working precision. This condition is indicated by a return code of $\text{INFO} = \text{N} + 1$.

13: FERR(NRHS) – REAL (KIND=nag_wp) array *Output*

On exit: the forward error bound for each solution vector $\hat{x}_j$ (the jth column of the solution matrix X). If x_j is the true solution corresponding to $\hat{x}_j$, $\text{FERR}(j)$ is an estimated upper bound for the magnitude of the largest element in $(\hat{x}_j - x_j)$ divided by the magnitude of the largest element in $\hat{x}_j$.

14: BERR(NRHS) – REAL (KIND=nag_wp) array *Output*

On exit: the component-wise relative backward error of each solution vector $\hat{x}_j$ (i.e., the smallest relative change in any element of A or B that makes $\hat{x}_j$ an exact solution).

15: $\text{WORK}(2 \times \text{N})$ – REAL (KIND=nag_wp) array *Workspace*

16: INFO – INTEGER *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$ and $\text{INFO} \leq \text{N}$

The leading minor of order $\langle \textit{value} \rangle$ of A is not positive definite, so the factorization could not be completed, and the solution has not been computed. $\text{RCOND} = 0.0$ is returned.

$\text{INFO} = \text{N} + 1$

D is nonsingular, but RCOND is less than ***machine precision***, meaning that the matrix is singular to working precision. Nevertheless, the solution and error bounds are computed because there are a number of situations where the computed solution can be more accurate than the value of RCOND would suggest.

7 Accuracy

For each right-hand side vector b, the computed solution $\hat{x}$ is the exact solution of a perturbed system of equations $(A + E)\hat{x} = b$, where

$$|E| \leq c(n)\epsilon|R|\left|R^{\mathrm{T}}\right|, \text{ where } R = LD^{\frac{1}{2}},$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***. See Section 10.1 of Higham (2002) for further details.

If x is the true solution, then the computed solution $\hat{x}$ satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|\hat{x}\|_\infty} \leq w_c \operatorname{cond}(A, \hat{x}, b)$$

where $\operatorname{cond}(A, \hat{x}, b) = \left\|\left|A^{-1}\right|(|A||\hat{x}| + |b|)\right\|_\infty / \|\hat{x}\|_\infty \leq \operatorname{cond}(A) = \left\|\left|A^{-1}\right||A|\right\|_\infty \leq \kappa_\infty(A)$. If $\hat{x}$ is the jth column of X, then w_c is returned in $\mathrm{BERR}(j)$ and a bound on $\|x - \hat{x}\|_\infty / \|\hat{x}\|_\infty$ is returned in $\mathrm{FERR}(j)$. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F07JBF (DPTSVX) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07JBF (DPTSVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The number of floating-point operations required for the factorization, and for the estimation of the condition number of A is proportional to n. The number of floating-point operations required for the solution of the equations, and for the estimation of the forward and backward error is proportional to nr, where r is the number of right-hand sides.

The condition estimation is based upon Equation (15.11) of Higham (2002). For further details of the error estimation, see Section 4.4 of Anderson *et al.* (1999).

The complex analogue of this routine is F07JPF (ZPTSVX).

10 Example

This example solves the equations

$$AX = B,$$

where A is the symmetric positive definite tridiagonal matrix

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 6.0 & 10.0 \\ 9.0 & 4.0 \\ 2.0 & 9.0 \\ 14.0 & 65.0 \\ 7.0 & 23.0 \end{pmatrix}.$$

Error estimates for the solutions and an estimate of the reciprocal of the condition number of A are also output.

10.1 Program Text

```
    Program f07jbfe

!     F07JBF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: dptsvx, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: rcond
      Integer                          :: i, ifail, info, ldb, ldx, n, nrhs
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: b(:,:), berr(:), d(:), df(:), e(:),  &
                                          ef(:), ferr(:), work(:), x(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F07JBF Example Program Results'
      Write (nout,*)
      Flush (nout)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      ldb = n
      ldx = n
      Allocate (b(ldb,nrhs),berr(nrhs),d(n),df(n),e(n-1),ef(n-1),ferr(nrhs), &
        work(2*n),x(ldx,nrhs))

!     Read the lower bidiagonal part of the tridiagonal matrix A and
!     the right hand side b from data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Solve the equations AX = B for X

!     The NAG name equivalent of dptsvx is f07jbf
      Call dptsvx('Not factored',n,nrhs,d,e,df,ef,b,ldb,x,ldx,rcond,ferr,berr, &
        work,info)

      If ((info==0) .Or. (info==n+1)) Then

!       Print solution, error bounds and condition number

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,nrhs,x,ldx,'Solution(s)',ifail)

        Write (nout,*)
        Write (nout,*) 'Backward errors (machine-dependent)'
        Write (nout,99999) berr(1:nrhs)
        Write (nout,*)
        Write (nout,*) 'Estimated forward error bounds (machine-dependent)'
```

```
        Write (nout,99999) ferr(1:nrhs)
        Write (nout,*)
        Write (nout,*) 'Estimate of reciprocal condition number'
        Write (nout,99999) rcond

        If (info==n+1) Then
          Write (nout,*)
          Write (nout,*) 'The matrix A is singular to working precision'
        End If
      Else
        Write (nout,99998) 'The leading minor of order ', info, &
          ' is not positive definite'
      End If

99999 Format (1X,1P,7E11.1)
99998 Format (1X,A,I3,A)
    End Program f07jbfe
```

10.2 Program Data

```
F07JBF Example Program Data
  5     2                     :Values of N and NRHS
  4.0  10.0  29.0  25.0   5.0 :End of diagonal D
 -2.0  -6.0  15.0   8.0       :End of sub-diagonal E
  6.0  10.0
  9.0   4.0
  2.0   9.0
 14.0  65.0
  7.0  23.0                   :End of matrix B
```

10.3 Program Results

```
 F07JBF Example Program Results

 Solution(s)
             1          2
 1      2.5000     2.0000
 2      2.0000    -1.0000
 3      1.0000    -3.0000
 4     -1.0000     6.0000
 5      3.0000    -5.0000

 Backward errors (machine-dependent)
     0.0E+00    7.4E-17

 Estimated forward error bounds (machine-dependent)
     2.4E-14    4.7E-14

 Estimate of reciprocal condition number
     9.5E-03
```